Singularity Analysis for the Perspective-Five-Line Problem

International Journal of Computer Vision

Jorge García Fontán, Abhilash Nayak, Sébastien Briot, Mohab Safey El Din

Jorge García Fontán: Sorbonne Université, Laboratoire d'Informatique de Paris 6, LIP6, Equipe PolSys, Paris, France

Abhilash Nayak: Centre national de la Recherche Scientifique, Laboratoire des Sciences du Numérique de Nantes (LS2N), UMR CNRS 6004, Nantes, France

Sébastien Briot: Centre national de la Recherche Scientifique, Laboratoire des Sciences du Numérique de Nantes (LS2N), UMR CNRS 6004, Nantes, France

Mohab Safey El Din: Sorbonne Université, CNRS, INRIA, Laboratoire d'Informatique de Paris 6, LIP6, Equipe PolSys, Paris, France

Corresponding author: Jorge García Fontán, email: Jorge.Garcia-Fontan@lip6.fr

1. Defining the lines, finding the interaction matrix and calssifying its minors into ideals G, H and K

```
> restart;
```

Calling the required libraries

```
> with(LinearAlgebra): with(plots): with(plottools):
```

Defining the 5 lines (direction vector and a point on the line) and the position of the camera in the object frame

```
> OC:=Vector([X,Y,Z]):
> OP1:=Vector([0,0,0]):  CP1:=OP1-OC: u1:=Vector([1,0,0]):
  OP2:=Vector([0,0,d1]): CP2:=OP2-OC: u2:=Vector([s1,s2,0]):
  OP3:=Vector([d2,d3,0]): CP3:=OP3-OC: u3:=Vector([d2,s3,s4]):
  OP4:=Vector([0,d4,d5]): CP4:=OP4-OC: u4:=Vector([s5,d4,s6]):
  OP5:=Vector([d6,0,d7]): CP5:=OP5-OC: u5:=Vector([s7,s8,d7]):
```

Plotting the observed lines for arbitrary parameters

```
> val:={d1=2,d2=3,d3=5,d4=1,d5=7,d6=-4,d7=11,  s1=4,s2=-5,s3=7,s4=
  3,s5=-2,s6=13,s7=-11,s8=6}:
> n:=1:
  t1:=spacecurve(eval(OP1,val)+lambda*eval(u1,val),lambda=-20..20,
  color=red,thickness=5,transparency=0):
  t2:=spacecurve(eval(OP2,val)+lambda*eval(u2,val),lambda=-n..n,
  color=green,thickness=5,transparency=0):
  t3:=spacecurve(eval(OP3,val)+lambda*eval(u3,val),lambda=-n..n,
  color=blue,thickness=5,transparency=0):
  t4:=spacecurve(eval(OP4,val)+lambda*eval(u4,val),lambda=-n..n,
  color=black,thickness=5,transparency=0):
  t5:=spacecurve(eval(OP5,val)+lambda*eval(u5,val),lambda=-n..n,
  color=cyan,thickness=5,transparency=0):
> display(t1,t2,t3,t4,t5,labels=[X,Y,Z],axes=normal);
```



The rows of the interaction matrix consisting of an affine line and an ideal line for each observed line

```
> for i from 1 to 5 do
  f||i||1:=CrossProduct(u||i,CP||i):
  m||i||1:=CrossProduct(OP||i,f||i||1):
  f||i||2:=ZeroVector(3):
  m||i||2:=CrossProduct(u||i,f||i||1):
  zf||i:=Vector([Transpose(f||i||1),Transpose(m||i||1)]):
  zm||i:=Vector([Transpose(f||i||2),Transpose(m||i||2)]):
  od:
> zeta:=Transpose(Matrix([zf1,zf2,zf3,zf4,zf5,zm1,zm2,zm3,zm4,zm5]))
  ) ;
```

$$\zeta := \Big[\big[0, Z, -Y, 0, 0, 0\big],$$ **(1.1)**

$$\big[s2\,(d1-Z),\ -s1\,(d1-Z),\ X\,s2-Y\,s1,\ d1\,s1\,(d1-Z),\ d1\,s2\,(d1-Z),\ 0\big],$$

$$[-s3\,Z-s4\,(d3-Y),\ d2\,Z+s4\,(d2-X),\ d2\,(d3-Y)-s3\,(d2-X),\ d3\,(d2\,(d3-Y)-s3\,(d2-X)),\ -d2\,(d2\,(d3-Y)-s3\,(d2-X)),\ d2\,(d2\,Z+s4\,(d2-X))-d3\,(-s3\,Z-s4\,(d3-Y))],$$

$$[d4\,(d5-Z)-s6\,(d4-Y),\ -s5\,(d5-Z)-s6\,X,\ s5\,(d4-Y)+d4\,X,\ d4\,(s5\,(d4-Y)+d4\,X)-d5\,(-s5\,(d5-Z)-s6\,X),\ d5\,(d4\,(d5-Z)-s6\,(d4-Y)),\ -d4\,(d4\,(d5-Z)-s6\,(d4-Y))],$$

$$[s8\,(d7-Z)+d7\,Y,\ -s7\,(d7-Z)+d7\,(d6-X),\ -s7\,Y-s8\,(d6-X),\ -d7\,(-s7\,(d7-Z)+d7\,(d6-X)),\ -d6\,(-s7\,Y-s8\,(d6-X))+d7\,(s8\,(d7-Z)+d7\,Y),\ d6\,(-s7\,(d7-Z)+d7\,(d6-X))],$$

$$\big[0, 0, 0, 0, Y, Z\big],$$

$$\big[0, 0, 0, s2\,(X\,s2-Y\,s1),\ -s1\,(X\,s2-Y\,s1),\ -s1^2\,(d1-Z)-s2^2\,(d1-Z)\big],$$

$[0, 0, 0, s3\,(d2\,(d3 - Y) - s3\,(d2 - X)) - s4\,(d2\,Z + s4\,(d2 - X)), -d2\,(d2\,(d3 - Y) - s3\,(d2 - X)) + s4\,(-s3\,Z - s4\,(d3 - Y)), d2\,(d2\,Z + s4\,(d2 - X)) - s3\,(-s3\,Z - s4\,(d3 - Y))],$

$[0, 0, 0, d4\,(s5\,(d4 - Y) + d4\,X) - s6\,(-s5\,(d5 - Z) - s6\,X), -s5\,(s5\,(d4 - Y) + d4\,X) + s6\,(d4\,(d5 - Z) - s6\,(d4 - Y)), s5\,(-s5\,(d5 - Z) - s6\,X) - d4\,(d4\,(d5 - Z) - s6\,(d4 - Y))],$

$[0, 0, 0, s8\,(-s7\,Y - s8\,(d6 - X)) - d7\,(-s7\,(d7 - Z) + d7\,(d6 - X)), -s7\,(-s7\,Y - s8\,(d6 - X)) + d7\,(s8\,(d7 - Z) + d7\,Y), s7\,(-s7\,(d7 - Z) + d7\,(d6 - X)) - s8\,(s8\,(d7 - Z) + d7\,Y)]]$

210 minors of the interaction matrix

```
> ch:=combinat[choose](10,4): nops(ch);
```

210 **(1.2)**

```
> for i from 1 to nops(ch) do
  zt||i:=DeleteRow(zeta,ch[i]):
  D||i:=factor(Determinant(DeleteRow(zeta,ch[i]))):
  od:
> I210:=[seq(D||i,i=1..210)]: nops(%);
```

210 **(1.3)**

Determining the ideals G, H and K

```
> zf:=Transpose(Matrix([zf1,zf2,zf3,zf4,zf5])):
  zm:=Transpose(Matrix([zm1,zm2,zm3,zm4,zm5])):
> c52:=combinat[choose](5,2); nc:=nops(c52):
```

$c52 := [[1, 2], [1, 3], [1, 4], [1, 5], [2, 3], [2, 4], [2, 5], [3, 4], [3, 5], [4, 5]]$ **(1.4)**

```
> k:=1: #splitting some minors into ideals G and H.
  for i from 1 to nc do
  for j from 1 to nc do
  zs:=Matrix([[DeleteRow(zf,c52[i])],[DeleteRow(zm,c52[j])]]):
  Ds||k:=factor(Determinant(zs)):
  Gs||k:=factor(Determinant(DeleteRow(zf[1..5,1..3],c52[i]))):
  Hs||k:=factor(Determinant(DeleteRow(zm[1..5,4..6],c52[j]))):
  k:=k+1:
  od:
  od:
  I16:=[seq(Ds||i,i=1..k-1)]:
  G16t:=[seq(Gs||i,i=1..k-1)]: ListTools[MakeUnique](G16t): G:=[seq
  (%[i],i=1..nops(%))]: nops(%);
  H16t:=[seq(Hs||i,i=1..k-1)]: ListTools[MakeUnique](H16t): H:=[seq
  (%[i],i=1..nops(%))]: nops(%);
```

10

10 **(1.5)**

The remaining 6 minors that cannot be factorized: Ideal K55

```
> k:=1:
  for i from 1 to 210 do
  if nops(I210[i])=2 or nops(I210[i])=3 then
  elif I210[i]=0 then
  pluch:
  else
  Kv[k]:=I210[i]:
  k:=k+1:
```

```
fi:
od:
k:=k-1:
> K55:=[seq(Kv[i],i=1..k)]: nops(%);
```

$$55 \tag{1.6}$$

Substituting parameters as given in Appendix A.

```
> Gv:=eval(G,val);
```

$$Gv := [7167 X^2 + 3796 X Y + 4339 X Z - 3630 Y^2 - 516 Y Z + 604 Z^2 - 9459 X - 678 Y - 3250 Z + 1074, -3950 X^2 - 6185 X Y - 2242 X Z - 2420 Y^2 - 1356 Y Z - 206 Z^2 + 9384 X + 6632 Y + 2910 Z - 4996, -2405 X^2 - 1924 X Y + 733 X Z + 2278 Y Z + 846 Z^2 + 10559 X + 5064 Y + 1554 Z - 6492, -1765 X^2 - 587 X Y - 878 X Z + 660 Y^2 + 232 Y Z - 122 Z^2 + 2606 X + 216 Y + 598 Z - 708, -924 X Y + 102 X Z - 847 Y^2 - 341 Y Z + 36 Z^2 + 1386 Y + 12 Z, -363 X Y + 552 X Z - 132 Y^2 + 347 Y Z + 204 Z^2 + 1749 Y - 36 Z, -177 X Y - 27 X Z + 75 Y^2 + 17 Y Z - 8 Z^2 + 156 Y + 6 Z, -110 X Y - 270 X Z - 88 Y^2 - 337 Y Z - 120 Z^2 + 242 Y + 240 Z, -130 X Y + 40 X Z - 104 Y^2 - 62 Y Z + 10 Z^2 + 188 Y - 20 Z, -30 X Y + 145 X Z - 24 Y^2 + 11 Y Z + 30 Z^2 + 210 Y - 60 Z] \tag{1.7}$$

```
> Hv:=eval(H,val);
```

$$Hv := [162602 X^3 + 46906 X^2 Y + 1612778 X^2 Z - 790636 X Y^2 + 1768668 X Y Z + 71038 X Z^2 + 390258 Y^3 - 794372 Y^2 Z - 1067620 Y Z^2 + 80122 Z^3 - 6124500 X^2 - 10320180 X Y + 3756084 X Z + 1357554 Y^2 + 3276582 Y Z + 1308192 Z^2 + 33121210 X + 23013522 Y - 6710602 Z - 34432464, -366610 X^3 + 303947 X^2 Y - 67779 X^2 Z + 256468 X Y^2 + 813154 X Y Z - 428220 X Z^2 - 177056 Y^3 + 73270 Y^2 Z - 660367 Y Z^2 - 5863 Z^3 + 2425148 X^2 - 3315971 X Y + 8642341 X Z - 2963608 Y^2 + 5422957 Y Z + 1597073 Z^2 - 15148022 X - 7865422 Y - 13002740 Z + 19664092, -217295 X^3 + 53924 X^2 Y + 35625 X^2 Z + 224118 X Y^2 + 53960 X Y Z + 239951 X Z^2 + 33528 Y^3 - 417430 Y^2 Z + 80572 Y Z^2 + 128535 Z^3 + 2141480 X^2 + 324074 X Y + 445690 X Z - 235692 Y^2 - 1754620 Y Z - 907002 Z^2 - 9471279 X - 2909124 Y - 2279067 Z + 7157862, -73960 X^3 - 46428 X^2 Y + 320426 X^2 Z + 88867 X Y^2 + 163934 X Y Z + 184389 X Z^2 + 62940 Y^3 - 356381 Y^2 Z - 32282 Y Z^2 + 27183 Z^3 + 210018 X^2 - 721747 X Y - 416981 X Z - 146898 Y^2 - 118097 Y Z - 116973 Z^2 + 111106 X + 377082 Y + 153504 Z - 56580, -16488 X^2 Y + 10906 X^2 Z + 10653 X Y^2 - 11926 X Y Z - 7705 X Z^2 - 726 Y^3 - 10775 Y^2 Z - 17392 Y Z^2 - 143 Z^3 + 211842 X Y + 170436 X Z - 1971 Y^2 + 139512 Y Z + 38667 Z^2 - 213246 Y - 239806 Z, -8431 X^2 Y + 7125 X^2 Z + 2478 X Y^2 + 11210 X Y Z + 1416 X Z^2 - 2772 Y^3 - 2280 Y^2 Z + 6847 Y Z^2 + 3135 Z^3 + 107494 X Y + 42342 X Z - 7182 Y^2 - 48208 Y Z - 15852 Z^2 - 178563 Y - 87291 Z, -3038 X^2 Y + 3686 X^2 Z - 2288 X Y^2 + 16544 X Y Z + 3344 X Z^2 - 315 Y^3 - 4111 Y^2 Z - 157 Y Z^2 + 663 Z^3 + 27166 X Y - 4168 X Z + 2769 Y^2 - 13942 Y Z - 1527 Z^2 - 25806 Y + 690 Z, \tag{1.8}$$

$$-3025\,X^2\,Y-1490\,X^2\,Z-770\,X\,Y^2+6050\,X\,Y\,Z-4070\,X\,Z^2+1320\,Y^3+3350\,Y^2\,Z+1705\,Y\,Z^2+18201\,X\,Y+38810\,X\,Z+19448\,Y^2-7697\,Y\,Z+57646\,Y,\ -650\,X^2\,Y-3350\,X^2\,Z-195\,X\,Y^2+8450\,X\,Y\,Z-845\,X\,Z^2+260\,Y^3+3410\,Y^2\,Z+390\,Y\,Z^2-13390\,X\,Y+1630\,X\,Z+276\,Y^2-8372\,Y\,Z+15088\,Y,\ 225\,X^2\,Y-1685\,X^2\,Z+705\,X\,Y^2+450\,X\,Y\,Z-345\,X\,Z^2+420\,Y^3-1325\,Y^2\,Z-645\,Y\,Z^2-8056\,X\,Y+705\,X\,Z-918\,Y^2-1527\,Y\,Z+5658\,Y]$$

```
> K55v:=eval(K55,val):
```

2. Analysis of the variety V(G) intersection V(K)

Groebner basis of G

```
> indets(G);
```

$$\{X, Y, Z, d1, d2, d3, d4, d5, d6, d7, s1, s2, s3, s4, s5, s6, s7, s8\} \quad (2.1)$$

```
> GbG:=Groebner[Basis](G,plex(d1, d2, d3, d4, d5, d6, d7, s1, s2, s3, s4, s5, s6, s7, s8, X, Y, Z)): nops(%);
```

$$10 \quad (2.2)$$

The normal form of elements of K w.r.t the Groebner basis.

```
> Groebner[NormalForm](K55[55],GbG,plex(d1, d2, d3, d4, d5, d6, d7, s1, s2, s3, s4, s5, s6, s7, s8, X, Y, Z)); # It can be proved for the rest of the elements of K55 but we know that Thus V(G) intersection V(K) = V(G)
```

$$0 \quad (2.3)$$

Groebner basis of G after substituting the parameters d_i and s_i

```
> GbGv:=Groebner[Basis](Gv,plex(X,Y,Z)); # No solution for generic five lines
```

$$GbGv := [1] \quad (2.4)$$

Finding five lines such that GbGv is not <1> leading to a singularity

```
> for i from 1 to 5 do
l||i:=CrossProduct(u||i,OP||i):
le||i:=Vector([Transpose(u||i),Transpose(l||i)]):
od:
> L5:=Transpose(Matrix([le1,le2,le3,le4,le5]));
```

$$L5 := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 \\ s1 & s2 & 0 & s2\,d1 & -s1\,d1 & 0 \\ d2 & s3 & s4 & -s4\,d3 & s4\,d2 & d2\,d3-d2\,s3 \\ s5 & d4 & s6 & d4\,d5-d4\,s6 & -s5\,d5 & s5\,d4 \\ s7 & s8 & d7 & s8\,d7 & d6\,d7-d7\,s7 & -s8\,d6 \end{bmatrix} \quad (2.5)$$

```
> Eqs:=[op(convert(L5[2..5,1..6].Vector([0,k1,k2,k3,k4,1]),list)), k1*k4+k2];
```

$$Eqs := [d1\,k3\,s2-d1\,k4\,s1+k1\,s2,\ d2\,k4\,s4-d3\,k3\,s4+d2\,d3-d2\,s3+k1\,s3+k2\,s4,\ d4\,k1+s6\,k2+(d4\,d5-d4\,s6)\,k3-s5\,d5\,k4+s5\,d4,\ s8\,k1+d7\,k2+s8\,d7\,k3+(d6\,d7-d7\,s7)\,k4-s8\,d6,\ k1\,k4+k2] \quad (2.6)$$

```
> El:=eliminate(Eqs,{k1,k2,k3,k4}):
> h:=op(El[2]): # If this polynomial is zero, we have a singularity
```

Example

```
> v2:={d1=2,d2=3,d3=5,d4=1,d5=7,d6=-4, s1=4,s2=-5,s3=7,s4=3,s5=-2,
  s6=13,s7=-11,s8=6}:
> sd7:=[solve(eval(h,v2))];
```

$$sd7 := \left[\frac{128893236}{7630285} + \frac{24}{7630285}\sqrt{8508173023861}, \frac{128893236}{7630285} - \frac{24}{7630285}\sqrt{8508173023861}\right] \quad (2.7)$$

```
> Gv2:=map(numer,eval(G,{op(v2),d7=sd7[1]})):
> L:=Groebner[Basis](Gv2,plex(X,Y,Z)); # V(L) is the line that
  intersects the five observed lines if they are chosen according
  to the parameters v2. Alternatively, if we make sure that the
  five lines do not satsify h=0, they belong to a regular linear
  line complex and we do not have one dimensional singularities.
```

$$L := \left[2166\,Y + \left(-\sqrt{448189} + 583\right) Z, 2166\sqrt{448189} - 2822298 + 3822990\,X + \left(329\sqrt{448189} + 587953\right) Z\right] \quad (2.8)$$

3. Analsysis of the variety V(H) intersection V(K)

Groebner basis of H after substituting the parameters d_i and s_i

```
> GbHv:=Groebner[Basis](Hv,plex(X, Y, Z)): nops(%);
```

$$3 \quad (3.1)$$

The normal form of elements of K w.r.t the Groebner basis.

```
> Groebner[NormalForm](K55v[1],GbHv,plex(X, Y, Z)): nops(%); # Non-
  zero for all 55 elements of K55.
```

$$10 \quad (3.2)$$

V(H) intersection V(K)

```
> HKv:=[op(Hv),op(K55v)]: indets(%);
```

$$\{X, Y, Z\} \quad (3.3)$$

```
> Groebner[Basis](HKv,plex(X,Y,Z)); # No solutions for 5 generic
  lines
```

$$[1] \quad (3.4)$$

Results : There are no singularities in P5P when the 5 observed lines are chosen generically. Some conditions on the relative configurations of the lines must be satisfied to have a one-dimensional singularity, a transversal that intersects the five lines. Finding those conditions for the existence of isolated singularities is an open problem.

4. Singularities in P5L when the lines are subjected to

orthogonality and parallelism constraints

Parametrization

```
> OP1:=Vector([0,0,0]): CP1:=OP1-OC: u1:=Vector([1,0,0]):
  OP2:=Vector([0,0,d1]): CP2:=OP2-OC: u2:=Vector([0,1,0]):
  OP3:=Vector([d2,d3,0]): CP3:=OP3-OC: u3:=Vector([0,0,1]):
  OP4:=Vector([0,d4,d5]): CP4:=OP4-OC: u4:=Vector([1,0,0]):
  OP5:=Vector([d6,0,d7]): CP5:=OP5-OC: u5:=Vector([0,1,0]):
> t1:=spacecurve(eval(OP1,val)+lambda*eval(u1,val),lambda=-20..20,
  color=red,thickness=5,transparency=0):
  t2:=spacecurve(eval(OP2,val)+lambda*eval(u2,val),lambda=-20..20,
  color=green,thickness=5,transparency=0):
  t3:=spacecurve(eval(OP3,val)+lambda*eval(u3,val),lambda=-20..20,
  color=blue,thickness=5,transparency=0):
  t4:=spacecurve(eval(OP4,val)+lambda*eval(u4,val),lambda=-20..20,
  color=black,thickness=5,transparency=0):
  t5:=spacecurve(eval(OP5,val)+lambda*eval(u5,val),lambda=-20..20,
  color=magenta,thickness=5,transparency=0):
> display(t1,t2,t3,t4,t5,labels=[X,Y,Z],axes=normal);
```



Interaction matrix and deriving ideals G, H and K55

```
> for i from 1 to 5 do
  f||i||1:=CrossProduct(u||i,CP||i):
  m||i||1:=CrossProduct(OP||i,f||i||1):
  f||i||2:=ZeroVector(3):
  m||i||2:=CrossProduct(u||i,f||i||1):
  zf||i:=Vector([Transpose(f||i||1),Transpose(m||i||1)]):
  zm||i:=Vector([Transpose(f||i||2),Transpose(m||i||2)]):
  od:
> zeta:=Transpose(Matrix([zf1,zf2,zf3,zf4,zf5,zm1,zm2,zm3,zm4,zm5])
  );
```

$$\zeta := [[0, Z, -Y, 0, 0, 0], \quad (4.1)$$
$$[d1 - Z, 0, X, 0, d1\,(d1 - Z), 0],$$
$$[-d3 + Y, d2 - X, 0, 0, 0, d2\,(d2 - X) - d3\,(-d3 + Y)],$$
$$[0, -d5 + Z, d4 - Y, d4\,(d4 - Y) - d5\,(-d5 + Z), 0, 0],$$
$$[d7 - Z, 0, -d6 + X, 0, -d6\,(-d6 + X) + d7\,(d7 - Z), 0],$$
$$[0, 0, 0, 0, Y, Z],$$

$[0, 0, 0, X, 0, -d1 + Z],$

$[0, 0, 0, -d2 + X, -d3 + Y, 0],$

$[0, 0, 0, 0, -d4 + Y, -d5 + Z],$

$[0, 0, 0, -d6 + X, 0, -d7 + Z]]$

```
> zf:=Transpose(Matrix([zf1,zf2,zf3,zf4,zf5])):
  zm:=Transpose(Matrix([zm1,zm2,zm3,zm4,zm5])):
```

```
> c52:=combinat[choose](5,2); nc:=nops(c52):
```

$$c52 := [[1, 2], [1, 3], [1, 4], [1, 5], [2, 3], [2, 4], [2, 5], [3, 4], [3, 5], [4, 5]] \tag{4.2}$$

```
> k:=1: #splitting some minors into two ideals G and H.
  for i from 1 to nc do
  for j from 1 to nc do
  zs:=Matrix([[DeleteRow(zf,c52[i])],[DeleteRow(zm,c52[j])]]):
  Ds||k:=factor(Determinant(zs)):
  Gs||k:=factor(Determinant(DeleteRow(zf[1..5,1..3],c52[i]))):
  Hs||k:=factor(Determinant(DeleteRow(zm[1..5,4..6],c52[j]))):
  k:=k+1:
  od:
  od:
  I16:=[seq(Ds||i,i=1..k-1)]:
  G16t:=[seq(Gs||i,i=1..k-1)]: ListTools[MakeUnique](G16t): G:=[seq
  (%[i],i=1..nops(%))]: nops(%);
  H16t:=[seq(Hs||i,i=1..k-1)]: ListTools[MakeUnique](H16t): H:=[seq
  (%[i],i=1..nops(%))]: nops(%);
```

$$10$$

$$10 \tag{4.3}$$

```
> ch:=combinat[choose](10,4): nops(ch):
```

```
> for i from 1 to nops(ch) do
  zt||i:=DeleteRow(zeta,ch[i]):
  D||i:=factor(Determinant(DeleteRow(zeta,ch[i]))):
  od:
```

```
> K55:=[seq(D||i,i=1..210)]:
```

Analysis of the variety V(G)

```
> Groebner[Basis](G,tdeg(X,Y,Z)); # No singularities in the generic
  case
```

$$[1] \tag{4.4}$$

Conditions such that V(G) is not null

```
> for i from 1 to 5 do
  l||i:=CrossProduct(u||i,OP||i):
  le||i:=Vector([Transpose(u||i),Transpose(l||i)]):
  od:
```

```
> L5:=Transpose(Matrix([le1,le2,le3,le4,le5]));
```

(4.5)

$$L5 := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & d1 & 0 & 0 \\ 0 & 0 & 1 & -d3 & d2 & 0 \\ 1 & 0 & 0 & 0 & -d5 & d4 \\ 0 & 1 & 0 & d7 & 0 & -d6 \end{bmatrix} \quad (4.5)$$

```
> flf:=1:
  for i from 1 to 5 do
  unassign('k1,k2,k3,k4,k5'):
  k||i:=1:
  Eqs:=[op(convert(L5[2..5,1..6].Vector([0,k1,k2,k3,k4,k5]),list)),
  k1*k4+k2*k5];
  El:=eliminate(Eqs,{k1,k2,k3,k4,k5} minus {k||i}):
  fl||i:=op(map(factor,El[2]));
  flf:=lcm(flf,fl||i):
  od:
> h:=flf;
```

$$h := (d1 - d7)\, d5\, (d1\, d2\, d4 - d1\, d4\, d6 - d2\, d4\, d7 + d3\, d5\, d6) \quad (4.6)$$

```
> Ga:=Groebner[Basis]([eval(op(G),d5=0)],plex(X,Y,Z)); nops(%); # If d5=0, the transversal is as follows.
```

$$Ga := [Z, -d2 + X]$$

$$2 \quad (4.7)$$

```
> Gb:=Groebner[Basis]([eval(op(G),d7=d1)],plex(X,Y,Z)); nops(%); # If d1-d7=0, the transversal is as follows.
```

$$Gb := [-d1 + Z, -d3 + Y]$$

$$2 \quad (4.8)$$

```
> Gct:=map(factor,Groebner[Basis]([op(G),d1*d2*d4-d1*d4*d6-d2*d4*d7+d3*d5*d6],plex(X,Y,Z,d1,d2,d3,d4,d5,d6,d7)));
```

$$Gct := [d1\, d2\, d4 - d1\, d4\, d6 - d2\, d4\, d7 + d3\, d5\, d6, d6\, d7\, (Y\, d5 - Z\, d4), d5\, d6\, (Y\, d5 - Z\, d4), (d1 - d7)\, (Y\, d5 - Z\, d4), (-d7 + Z)\, (Y\, d5 - Z\, d4), Z\, (Y\, d1\, d2 - Y\, d1\, d6 - Y\, d2\, d7 + Z\, d3\, d6), (-d3 + Y)\, (Y\, d5 - Z\, d4), d6\, (X\, d3\, d5 - X\, d4\, d7 - Y\, d2\, d5 + Y\, d5\, d6 + d2\, d4\, d7 - d3\, d5\, d6), d5\, (X\, d1 - X\, d7 + Z\, d6 - d1\, d6), X\, d1\, d4 - X\, d4\, d7 + Y\, d5\, d6 - d1\, d4\, d6, (-d7 + Z)\, (X\, d3\, d5 - X\, d4\, d7 - Z\, d2\, d4 + Z\, d4\, d6 + d2\, d4\, d7 - d3\, d5\, d6), Z\, (X\, d1 - X\, d7 + Z\, d6 - d1\, d6), X\, Y\, d7 - X\, Z\, d3 + Y\, Z\, d2 - Y\, Z\, d6 - Y\, d2\, d7 + Z\, d3\, d6, X\, Y\, d5 - X\, Z\, d4 - X\, d3\, d5 + X\, d4\, d7 - Y\, d5\, d6 + Z\, d2\, d4 - d2\, d4\, d7 + d3\, d5\, d6, X\, Y\, d1 - X\, Z\, d3 + Y\, Z\, d2 - Y\, d1\, d2, (-d2 + X)\, (X\, d1 - X\, d7 + Z\, d6 - d1\, d6)] \quad (4.9)$$

```
> map(factor,eval(Gct,{Y=Z*d4/d5}));
```

$$\Big[d1\, d2\, d4 - d1\, d4\, d6 - d2\, d4\, d7 + d3\, d5\, d6, 0, 0, 0, 0, \frac{Z^2\, (d1\, d2\, d4 - d1\, d4\, d6 - d2\, d4\, d7 + d3\, d5\, d6)}{d5}, 0, d6\, (X\, d3\, d5 - X\, d4\, d7 - Z\, d2\, d4 + Z\, d4\, d6 + d2\, d4\, d7 - d3\, d5\, d6), d5\, (X\, d1 - X\, d7 + Z\, d6 - d1\, d6), d4\, (X\, d1 - X\, d7 + Z\, d6 - d1\, d6), (-d7 + Z)\, (X\, d3\, d5 - X\, d4\, d7 - Z\, d2\, d4 + Z\, d4\, d6 + d2\, d4\, d7 \quad (4.10)$$

$-d3\,d5\,d6),\ Z\,(X\,d1-X\,d7+Z\,d6-d1\,d6),$

$-\frac{Z\,(X\,d3\,d5-X\,d4\,d7-Z\,d2\,d4+Z\,d4\,d6+d2\,d4\,d7-d3\,d5\,d6)}{d5},\ -X\,d3\,d5+X\,d4\,d7$

$+Z\,d2\,d4-Z\,d4\,d6-d2\,d4\,d7+d3\,d5\,d6,$

$\frac{Z\,(X\,d1\,d4-X\,d3\,d5+Z\,d2\,d4-d1\,d2\,d4)}{d5},\ (-d2+X)\,(X\,d1-X\,d7+Z\,d6-d1\,d6)\Big]$

```
> Gc:=[Y*d5-Z*d4, X*d1-X*d7+Z*d6-d1*d6]; # If the third factor of h vanishes, the transversal is as follows.
```

$$Gc := [Y\,d5-Z\,d4,\ X\,d1-X\,d7+Z\,d6-d1\,d6] \quad (4.11)$$

Analysis of the variety V(H) intersection V(K)

```
> Groebner[Basis]([op(H),op(K)],plex(X,Y,Z)); # No singularities in the generic case
```

$$[1] \quad (4.12)$$

Conditions such that V(H) intersection V(K) is not null

```
> GbH:=Groebner[Basis](H,plex(X,Y,Z)); nops(%);
```

$$GbH := [2\,Z\,d4\,d6+d1\,d2\,d4-d1\,d4\,d6-d2\,d4\,d7-d3\,d5\,d6,\ 2\,Y\,d5\,d6+d1\,d2\,d4-d1\,d4\,d6-d2\,d4\,d7-d3\,d5\,d6,\ X\,(2\,d1\,d4-2\,d4\,d7)-d1\,d2\,d4-d1\,d4\,d6+d2\,d4\,d7+d3\,d5\,d6]$$

$$3 \quad (4.13)$$

```
> Cs:=solve(GbH,{X,Y,Z}); # A point. It should belong to V(K) to have V(H) intersection V(K) = V(K) which is not null.
```

$$Cs := \left\{X=\frac{1}{2}\,\frac{d1\,d2\,d4+d1\,d4\,d6-d2\,d4\,d7-d3\,d5\,d6}{(d1-d7)\,d4},\ Y=-\frac{1}{2}\,\frac{d1\,d2\,d4-d1\,d4\,d6-d2\,d4\,d7-d3\,d5\,d6}{d5\,d6},\ Z=-\frac{1}{2}\,\frac{d1\,d2\,d4-d1\,d4\,d6-d2\,d4\,d7-d3\,d5\,d6}{d4\,d6}\right\} \quad (4.14)$$

```
> f:=map(factor,remove(has,map(numer,eval(K55,Cs)),0)): nops(%); indets(%%); # There are 36 elements which should vanish simultaneously for an isolated singularity to exist.
```

$$36$$

$$\{d1,\ d2,\ d3,\ d4,\ d5,\ d6,\ d7\} \quad (4.15)$$

Results: For the special case of P5L with orthogonality and parallelism constraints, generically, there are no singularities. The singularities appear as a line and/or as a point for some special relative configurations of the five lines.